Matematica in laboratorio

1. La risoluzione approssimata delle equazioni

Usiamo Excel

In questo paragrafo ti proponiamo alcune esercitazioni con Excel che riguardano gli argomenti proposti in questo capitolo. Vedremo come, con l'aiuto di un foglio elettronico, sia possibile applicare i metodi iterativi che abbiamo visto per risolvere equazioni in modo approssimato.

I esercitazione: il metodo di bisezione

Ti sarai sicuramente reso conto che gli algoritmi iterativi che abbiamo visto per la risoluzione approssimata delle equazioni sono piuttosto laboriosi nei calcoli anche utilizzando una calcolatrice tascabile. Possiamo però farci aiutare anche in questo caso da Excel. Troviamo le radici approssimate dell'equazione $x^3 - 3x^2 + 3 = 0$.

Dal grafico dell'equazione ad essa associata (***figura a lato***) troviamo che l'equazione ha tre radici distinte:



$$r_1 \in (-1, 0) \qquad r_2 \in (1, 2) \qquad r_3 \in (2, 3)$$

Nel foglio di lavoro che puoi vedere di seguito, e che riporta il calcolo di r_1, dobbiamo costruire le colonne con i valori degli estremi a e b degli intervalli che contengono la radice, il valore del punto medio m, la valutazione della funzione in tali punti; abbiamo poi aggiunto una colonna con il numero delle iterazioni dell'algoritmo e una con l'ampiezza dell'intervallo che contiene la soluzione.
Ti diamo le indicazioni per costruire la parte significativa di questa tabella.

B2:	inserisci il primo valore di a	(-1 per la prima soluzione, 1 per la seconda, 2 per la terza)
C2:	inserisci il primo valore di b	(0 per la prima soluzione, 2 per la seconda, 3 per la terza)
D2:	=(B2+C2)/2	(calcolo di $m = \dfrac{a+b}{2}$)
E2:	=B2^3−3*B2^2+3	(calcolo di $f(a)$)
F2:	=C2^3−3*C2^2+3	(calcolo di $f(b)$)
G2:	=D2^3−3*D2^2+3	(calcolo di $f(m)$)
H2:	=C2−B2	(calcolo dell'ampiezza dell'intervallo considerato)
B3:	=SE(E2*G2< 0;B2;D2)	(se $f(a)f(m) < 0$ mettiamo il valore a in B3 come primo numero dell'intervallo altrimenti mettiamo il valore m)
C3:	=SE(F2*G2< 0;C2;D2)	(se $f(b)f(m) < 0$ mettiamo il valore b in C3 come secondo numero dell'intervallo, altrimenti mettiamo il valore m)

Con queste due ultime istruzioni abbiamo detto al computer di considerare l'intervallo (a, m) se $f(a) \cdot f(m) < 0$, oppure l'intervallo (m, b) se $f(b) \cdot f(m) < 0$.

Per iterare il procedimento, copia ora le formule per tutta la colonna cui si riferiscono, per esempio fino alla riga 13.

In questo modo abbiamo previsto di fare 12 iterazioni; se vuoi una precisione migliore puoi aumentare il numero di iterazioni copiando le celle fino all'iterazione desiderata.
I valori delle colonne B e C rappresentano le successioni dei valori approssimati per difetto e per eccesso della soluzione cercata; l'approssimazione migliora al crescere del numero di iterazioni.

La tabella che segue riguarda la soluzione che appartiene all'intervallo $(-1,0)$. Puoi trovare da solo la seconda soluzione che appartiene all'intervallo $(1,2)$ e la terza soluzione che appartiene all'intervallo $(2,3)$.

	A	B	C	D	E	F	G	H
1	iterazione	*a*	*b*	*m*	*f(a)*	*f(b)*	*f(m)*	ampiezza
2	1	-1,000000	0,000000	-0,500000	-1,000000	3,000000	2,125000	1,000000
3	2	-1,000000	-0,500000	-0,750000	-1,000000	2,125000	0,890625	0,500000
4	3	-1,000000	-0,750000	-0,875000	-1,000000	0,890625	0,033203	0,250000
5	4	-1,000000	-0,875000	-0,937500	-1,000000	0,033203	-0,460693	0,125000
6	5	-0,937500	-0,875000	-0,906250	-0,460693	0,033203	-0,208160	0,062500
7	6	-0,906250	-0,875000	-0,890625	-0,208160	0,033203	-0,086094	0,031250
8	7	-0,890625	-0,875000	-0,882813	-0,086094	0,033203	-0,026101	0,015625
9	8	-0,882813	-0,875000	-0,878906	-0,026101	0,033203	-0,003637	0,007813
10	9	-0,882813	-0,878906	-0,880859	-0,026101	0,003637	-0,011210	0,003906
11	10	-0,880859	-0,878906	-0,879883	-0,011210	0,003637	-0,003781	0,001953
12	11	-0,879883	-0,878906	-0,879395	-0,003781	0,003637	-0,000071	0,000977
13	12	-0,879395	-0,878906	-0,879150	-0,000071	0,003637	0,001784	0,000488

II esercitazione: il metodo delle secanti

Troviamo le radici approssimate dell'equazione $e^{x^2-1} - x = 0$ già vista nell'esempio del paragrafo corrispondente.

B3: 0 (estremo sinistro dell'intervallo $[a,b]$)

D3: 0,5 (estremo destro dell'intervallo $[a,b]$)

Nelle celle da A5 ad A17 costruisci la successione dei simboli c_i (inserisci le stringhe c0 e c1 nelle prime due celle, selezionale e poi trascina la selezione col mouse).
Il punto di partenza del metodo è il punto $b = 0,5$, inseriamo allora tale valore nella cella B5.

B5: =D3 ($c_0 = b$)

B6: =B5-((EXP(B5^2-1)-B5)*(B3-B5)) / ((EXP(B3^2-1)-B3)-(EXP(B5^2-1)-B5)) (formula per il calcolo di c_n)

Copia ora la formula della cella B6 da B7 a B17.
Puoi osservare che i valori trovati della radice, approssimati alla settima cifra decimale, si stabilizzano a partire dall'elemento c_{11}.

	A	B	C	D
1	IL METODO DELLE SECANTI			
2				
3	a=	**0**	b=	**0,5**
4				
5	c0	0,5		
6	c1	0,465066313		
7	c2	0,454733368		
8	c3	0,451851257		
9	c4	0,451060566		
10	c5	0,450844629		
11	c6	0,450785731		
12	c7	0,450769672		
13	c8	0,450765293		
14	c9	0,450764099		
15	c10	0,450763774		
16	c11	0,450763685		
17	c12	0,450763661		

III esercitazione: il metodo delle tangenti

Troviamo con questo metodo le radici approssimate della stessa equazione. Il punto di partenza è però questa volta il punto $a = 0$.
Ti diamo indicazioni solo sulle formule da costruire.

B5: =B3 ($d_0 = a$)

B6: =B5-((EXP(B5^2-1)-B5) / (2*B5*EXP(B5^2-1)-1)) (formula per il calcolo di d_n)

Copia ora la formula della cella B6 da B7 a B18.

In questo caso i valori approssimati trovati si stabilizzano già a partire da d_5.

	A	B	C	D
1	IL METODO DELLE TANGENTI			
2				
3	a=	**0**	b=	**0,5**
4				
5	d0	0		
6	d1	0,367879441		
7	d2	0,445133493		
8	d3	0,450730472		
9	d4	0,450763651		
10	d5	0,450763652		
11	d6	0,450763652		
12	d7	0,450763652		
13	d8	0,450763652		
14	d9	0,450763652		
15	d10	0,450763652		
16	d11	0,450763652		
17	d12	0,450763652		
18	d13	0,450763652		

IV esercitazione: il metodo del punto unito

Risolviamo l'equazione $2x - \cos^2 x = 0$ già vista nell'esempio del paragrafo corrispondente.

Costruisci, nella colonna A, a partire da A4 fino ad A18, la successione dei numeri naturali da 0 a 14 che rappresentano i valori di n. Nella colonna B costruiremo i valori della successione $\{x_n\}$ e nella colonna C i valori di $g(x_n)$.

B4: 0 (valore di x_0, punto iniziale scelto)

C4: =((cos(B4))^2) / 2 (calcolo di $g(x_0)$)

B5: =C4 ($x_1 = g(x_0)$)

Copia ora le formule della cella B5 da B6 a B18 e della cella C4 da C5 a C18.

Anche in questo caso puoi vedere che i valori di $g(x_n)$ si stabilizzano abbastanza in fretta.

	A	B	C	D
1	IL METODO DEL PUNTO UNITO			
2				
3		xi	g(xi)	
4	0	0	0,5	
5	1	0,5	0,385075576	
6	2	0,385075576	0,42945136	
7	3	0,42945136	0,41331716	
8	4	0,41331716	0,419338893	
9	5	0,419338893	0,417111705	
10	6	0,417111705	0,417938291	
11	7	0,417938291	0,417631904	
12	8	0,417631904	0,417745525	
13	9	0,417745525	0,417703397	
14	10	0,417703397	0,417719018	
15	11	0,417719018	0,417713226	
16	12	0,417713226	0,417715374	
17	13	0,417715374	0,417714577	
18	14	0,417714577	0,417714873	

Usiamo Wiris

Il comando per risolvere un'equazione trovando le sue soluzioni approssimate è il seguente

risolvere_numericamente (equazione)

E' poi anche possibile scegliere il metodo di risoluzione tra quello di bisezione e di Newton. Osserva i comandi elencati nella figura al termine dell'esercitazione; in particolare:

- per fissare il numero di cifre decimali si deve usare il comando
  precisione (n)
- per stabilire il metodo lo si deve indicare all'interno di una coppia di parentesi graffe mettendolo tra virgolette, precisando il punto di partenza nel caso del metodo di Newton, l'intervallo nel caso del metodo di bisezione; per esempio:

  {**metodo = "newton",punto_iniziale = 0.5002**}

  {**metodo = "bisezione", punto_iniziale = {2,3}**}